

# The Logic of Procedural Readings

①

Silvija Seres    Michael Spivey    Tony Hoare

*Oxford University Computing Laboratory  
Wolfson Building, Parks Road, Oxford OX1 3QD, U.K.*

## Abstract

The distinctive merit of the declarative reading of logic programs is the validity of all the laws of reasoning supplied by the predicate calculus with equality. Surprisingly many of these laws are still valid for the procedural reading; they can therefore be used safely for algebraic manipulation, program transformation and optimisation of executable logic programs.

This paper lists a number of common laws, and proves their validity for the standard (depth-first search) procedural reading of PROLOG. They also hold for alternative search strategies, e.g. breadth-first search. Our proofs of the laws are based on the standard algebra of functional programming, after the strategies have been given a rather simple implementation in a standard functional language.

## 1 Introduction

Logic programming languages are traditionally explained in terms of their declarative and procedural semantics. For a given logic program, they are respectively considered its specification and its model of execution. It is regarded as the responsibility of the programmer to ensure the consistency of the two readings. This paper aims to help in this, by codifying algebraic laws which apply equally to both readings. In some sense, a sufficient collection of these laws would provide an additional algebraic semantics for a logic language, intermediate between the declarative and procedural semantics. The general role of algebra in bridging the gap between abstract and concrete theories is argued in [8].

A proof of the validity of our laws in the declarative reading is unnecessary, because they express properties of Boolean algebra. To prove that they are true of the procedural reading requires us to construct a model of execution. This we do by implementing the operators of the logic language as a library of higher order functions in the functional language Haskell. This makes available all the algebraic reasoning principles of functional programming [3], from which it is quite straightforward to derive the laws we need. Many of these are familiar properties of categorical monads, but a knowledge of category theory is not needed for an understanding of this paper.

It is worth stressing that our implementation is a *shallow* embedding of a logic language in a functional one; it is *not* the same as building an interpreter. We do not extend the base functional language; rather, we implement *in* the language a set of functions designed to support unification, resolution and search.

Our implementation is strikingly simple, and the basic ideas that it builds upon are not new. The embedding of a logic language to a functional one by translating every predicate to a function was explored in e.g. LOGLISP [14, 15] or POPLOG [11], although the base language was non-lazy in each case. The use of the lazy stream-based execution model to compute the possibly infinite set of answers is also well known, e.g. [1]. Nevertheless, we believe that the combination of these two known ideas is well worth our attention, and the *algebraic* semantics for logic programs that naturally arises from our embedding is a convincing example. *confirmation*

To some extent, use of our library of functions will give functional programmer a small taste of the power of a *functional logic language*. But current functional logic languages are much more powerful; they embody both rewriting and resolution and thereby result in a functional language with the capability to solve arbitrary constraints for the values of variables. The list of languages that have been proposed in an attempt to incorporate the expressive power of both functional and logic paradigms is long and impressive [2, 6]; some notable examples are Kernel-LEAF [5], Curry [7], Escher [9] and Babel [12]. Our research goal is different from the one set by these projects. They aspire to build an efficient language that can offer programmers the most useful features of both worlds; to achieve this additional expressivity they have to adopt somewhat complicated semantics. Our present goal is a conspicuous declarative and operational semantics for the embedding, rather than maximal expressivity. Nevertheless, the extensions of our embedding to incorporate both narrowing and residuation in its operational semantics do not seem difficult and we hope to make them a subject of our further work. *per*

In this paper we use Prolog and Haskell as our languages of choice, but the principles presented are general. Prolog is chosen because it is the dominant logic language, although we only implement the pure declarative features of it, i.e., we ignore the impure but practically much used features like `cut`, `assert` and `retract` (although the `cut` is quite an easy extension of our models). Haskell is chosen because it is a lazy functional language with types and lambda-abstractions, but any other language with these properties could be used.

In the remainder of the paper we proceed to describe the syntax of the embedding and the implementation of the primitives in sections 2 and 3. In section 4 we list some of the algebraic properties of the operators and in section 5 we study the necessary changes to the system to accommodate different search strategies. We conclude the paper with section 6 where we discuss related work and propose some further work in this setting.

## 2 Syntax

Prolog offers the facility of defining a predicate in many clauses and it allows the applicability of each clause to be tested by pattern matching on the formal parameter list. In our implementation of Prolog, we have to withdraw these notational licences, and require the full logical meaning of the predicate to be defined in a single equation, with the unifications made explicit on the right hand side, together with the implicit existential quantification over the fresh variables.

In the proposed embedding of Prolog into a functional language, we aim to give rules that allow any pure Prolog predicate to be translated into a Haskell function with the same meaning. To this end, we introduce two data types, *Term* and *Predicate*, into our functional language, together with the following four operations:

$$\begin{aligned} (&), (||) : \textit{Predicate} \longrightarrow \textit{Predicate} \longrightarrow \textit{Predicate}, \\ (\doteq) : \textit{Term} \longrightarrow \textit{Term} \longrightarrow \textit{Predicate}, \\ \textit{exists} : (\textit{Term} \longrightarrow \textit{Predicate}) \longrightarrow \textit{Predicate}. \end{aligned}$$

The intention is that the operators  $\&$  and  $\parallel$  denote conjunction and disjunction of predicates,  $\doteq$  forms a predicate expressing the equality of two terms, and the operation *exists* expresses existential quantification. We shall abbreviate the expression *exists*  $(\lambda x \rightarrow p\ x)$  by the form  $\exists x \rightarrow p\ x$  in this paper, although the longer form shows how the expression can be written in any lazy functional language that has  $\lambda$ -expressions. We shall also write  $\exists x, y \rightarrow p(x, y)$  for  $\exists x \rightarrow (\exists y \rightarrow p(x, y))$ .

These four operations suffice to translate any pure Prolog program, provided we are prepared to exchange pattern matching for explicit equations, to bind local variables with explicit quantifiers, and to gather all the clauses defining a predicate into a single equation. These steps can be carried out systematically, and could easily be automated. As an example, we take the well-known program for append:

*append*  $([a|b], [c], [a|b|c])$  *true*

```
append([], Ys, Ys) :- .
append([X|Xs], Ys, [X|Zs]) :- append(Xs, Ys, Zs).
```

As a first step, we remove any patterns and repeated variables from the head of each clause, replacing them by explicit equations written at the start of the body. These equations are computed by unification in Prolog.

```
append(Ps, Qs, Rs) :-
  Ps = [], Qs = Rs.
append(Ps, Qs, Rs) :-
  Ps = [X|Xs], Rs = [X|Ys], append(Xs, Qs, Ys).
```

The head of each clause now contains only a list of distinct variables, and by renaming if necessary we can ensure that the list of variables is the same in each clause. We complete the translation by joining the clause bodies with the  $\parallel$  operation, the literals in a clause with the  $\&$  operation, and existentially quantifying any variables that appear in the body but not in the head of a clause:

$$\begin{aligned} \text{append}(Ps, Qs, Rs) = & \\ & (Ps \doteq \text{nil} \ \& \ Qs \doteq Rs) \parallel \\ & (\exists X, Xs, Ys \rightarrow Ps \doteq \text{cons}(X, Xs) \ \& \ Rs \doteq \text{cons}(X, Ys) \ \& \\ & \quad \text{append}(Xs, Qs, Ys) ). \end{aligned}$$

Here *nil* is used for the value of type *Term* that represents the empty list, and *cons* is written for the function on terms that corresponds to the Prolog list constructor `[]`. We assume the following order of precedence on the operators, from highest to lowest:  $\doteq, \&, \parallel, \exists$ .

The function *append* defined by this recursive equation has the following type:

$$\text{append} :: (\text{Term}, \text{Term}, \text{Term}) \longrightarrow \text{Predicate}.$$

The Haskell function *append* is constructed by making the *declarative* reading of the Prolog predicate explicit. However, the relationship between the Haskell function and the Prolog predicate extends beyond their declarative semantics. The next section shows that the *procedural* reading of the Prolog predicate is also preserved through the implementation of the functions  $\&$  and  $\parallel$ . The embedding essentially allows the mapping of the computation of the Prolog program into lazy lists by embedding the structure of a SLD-tree of a Prolog program into a Haskell stream.

### 3 Implementation

The translation described above depends on the four operations  $\&$ ,  $\parallel$ ,  $\doteq$  and *exists*. We now give definitions to the type of predicates and to these four operations that correspond to the depth-first search of Prolog. Later, we shall be able to give alternative definitions that correspond to breadth-first search, or other search strategies based on the search tree of the program.

The key idea is that each predicate is a function that takes an ‘answer’, representing the state of knowledge about the values of variables at the time the predicate is solved, and produces a lazy stream of answers, each corresponding to a solution of the predicate that is consistent with the input. This approach is similar to that taken by Wadler [18]. An unsatisfiable query results in an empty

stream, and a query with infinitely many answers results in an infinite stream.<sup>1</sup>

```
type Predicate = Answer -> Stream Answer.
```

An answer is (in principle) just a substitution, but we augment the substitution with a counter that tracks the number of variables that have been used so far, so that a fresh variable can be generated at any stage by incrementing the counter. A substitution is represented as a list of (variable, term) pairs, and the Haskell data-type *Term* is a straightforward implementation of Prolog's term type:

```
type Answer = (Subst, Int),
type Subst = [(Var, Term)],
data Term = Func Fname [Term] | Var Vname,
type Fname = String,
data Vname = Name String | Auto Int.
```

Constants are functions with arity 0, in other words they are given empty argument lists. For example the Prolog list [a,b] can be represented in the embedding as *Func "cons" [Func "a" [], Func "cons" [...]]*. With the use of the simple auxiliary functions *cons*, *atom* and *nil* the same Prolog list can be embedded as the Haskell expression *cons a (cons b nil)*.

We can now give definitions for the four operators. The operators & and || act as predicate combinators; they slightly resemble the notion of *tacticals* [13], but in our case they combine the computed streams of answers, rather than partially proved statements.

The || operator simply concatenates the streams of answers returned by its two operands:

```
(||) :: Predicate -> Predicate -> Predicate
(p || q) x = p x ++ q x.
```

This definition implies that the answers are returned in a left-to-right order as in Prolog. If the left-hand argument of || is unsuccessful and returns an empty answer stream, it corresponds to an unsuccessful branch of the search tree in Prolog and backtracking is simulated by evaluating the right-hand argument.

For the & operator, we start with applying the first argument to the incoming answer; this produces a stream of answers, to each of which we apply the second argument of &. Finally, we concatenate the resulting stream of streams into a single stream:

```
(&) :: Predicate -> Predicate -> Predicate
p & q = concat . map q . p.
```

---

<sup>1</sup>For clarity, we use the type constructor *Stream* to denote possibly infinite streams, and *List* to denote finite lists. In a lazy functional language, these two concepts share the same implementation.

Because of Haskell's lazy evaluation, the function  $p$  returns answers only when they are needed by the function  $q$ . This corresponds nicely with the backtracking behaviour of Prolog, where the predicate  $p \ \& \ q$  is implemented by enumerating the answers of  $p$  one at a time and filtering them with the predicate  $q$ . Infinite list of answers in Prolog are again modelled gracefully with infinite streams.

We can also define primitive predicates *true* and *false*, one corresponding to immediate success and the other to immediate failure:

$$\begin{array}{ll} \text{true} :: \text{Predicate} & \text{false} :: \text{Predicate} \\ \text{true } x = [x]. & \text{false } x = []. \end{array}$$

The pattern matching of Prolog is implemented by the operator  $\dot{=}$ . It is defined in terms of a function *unify* which implements J.A. Robinson's standard algorithm for s unification of two terms relative to a given input substitution. The type of *unify* is thus:

$$\text{unify} :: \text{Subst} \longrightarrow (\text{Term}, \text{Term}) \longrightarrow \text{List Subst.}$$

More precisely, the result of *unify*  $s (t, u)$  is either  $[s \triangleright r]$ , where  $r$  is a most general unifier of  $t[s]$  and  $u[s]$ , or  $[]$  if these two terms have no unifier.<sup>2</sup> Thus if *unify*  $s (t, u) = [s']$ , then  $s'$  is the most general substitution such that  $s \sqsubseteq s'$  and  $t[s'] = u[s']$ . The coding is routine and therefore omitted.

The  $\dot{=}$  operator is just a wrapper around *unify* that passes on the counter for fresh variables:

$$\begin{array}{l} (\dot{=}) :: \text{Term} \longrightarrow \text{Term} \longrightarrow \text{Predicate} \\ (t \dot{=} u) (s, n) = [(s', n) \mid s' \leftarrow \text{unify } s (t, u)] \end{array}$$

Finally, the operator *exists* is responsible for allocating fresh names for all the local (or existentially quantified) variables in the predicates. This is necessary in order to guarantee that the computed answer is the most general result. It is defined as follows:

$$\begin{array}{l} \text{exists} :: (\text{Term} \longrightarrow \text{Predicate}) \longrightarrow \text{Predicate} \\ \text{exists } p (s, n) = p (\text{makevar } n) (s, n + 1), \end{array}$$

where *makevar*  $n$  is a term representing the  $n$ 'th generated variable. The slightly convoluted flow of information here may be clarified by a small example. The argument  $p$  of *exists* will be a function that expects a variable, such as  $(\lambda X \rightarrow \text{append}(t, X, u))$ . We apply this function to a newly-invented variable  $v = \text{makevar } n$  to obtain the predicate *append*( $t, v, u$ ), and finally apply this predicate to the answer  $(s, n+1)$ , in which all variables up to the  $n$ 'th are marked as having been used.

<sup>2</sup>We use  $s \triangleright r$  to denote composition of substitutions  $s$  and  $r$ , and  $t[s]$  to denote the instance of term  $t$  under substitution  $s$ . We use  $s \sqsubseteq s'$  to denote the preorder on substitutions that holds iff  $s' = s \triangleright r$  for some substitution  $r$ .

The function *solve* evaluates the main query. It simply applies its argument, the predicate of the query, to an answer with an empty substitution and a zero variable counter, and converts the resulting stream of answers to a stream of strings.

$$\begin{aligned} \textit{solve} &:: \textit{Predicate} \longrightarrow \textit{Stream String} \\ \textit{solve } p &= \textit{map print } (p ([ ], 0)), \end{aligned}$$

where *print* is a function that converts an answer to a string by having pruned it to show only the values of the original query variables. This is the point where all the internally generated variables are filtered out in our present implementation, but another, possibly cleaner, solution might be to let the  $\exists$  operator do this filtering task before it returns.

We do not provide proofs of the soundness and completeness relative to the procedural reading of Prolog since we feel that the encoding we have described is about the simplest possible mechanised formal definition of a Prolog-like procedural reading. Nevertheless, a soundness proof for the embedding could be carried out relative to the declarative semantics by defining of a mapping *decl* between our embedding and a declarative semantics of a logic program. Given a function *herb* with type *Answer*  $\longrightarrow$  *Set Subst*:

$$\textit{herb}(s, n) = \{s; t \mid t \in \textit{Subst}\},$$

where *herb*(*s*, *n*) describes a set of all substitutions that refine (i.e. extend) the substitution part *s* of the input answer (*s*, *n*). The mapping from our embedding to the declarative semantics can then be defined as:

$$\textit{decl} = (\textit{fold union}) (\textit{map herb}).$$

Namely, if *P* is a predicate then *decl* · *P* is its declarative semantics. A soundness proof for the embedding would then be obtained by proving the equations:

$$\begin{aligned} \textit{decl} \cdot (P \parallel Q) &\subseteq (\textit{decl} \cdot P) \cup (\textit{decl} \cdot Q), \\ \textit{decl} \cdot (P \& Q) &\subseteq (\textit{decl} \cdot P) \cap (\textit{decl} \cdot Q), \end{aligned}$$

for the operators  $\parallel$  and  $\&$ , and similar equations for the operators  $\exists$  and  $\doteq$ .

## 4 Algebraic Laws

The operators  $\&$  and  $\parallel$  enjoy many algebraic properties as a consequence of their simple definitions in terms of streams. We can deduce directly from the implementation of  $\&$  and *true* that the  $\&$  operator is associative with unit element *true*. This is a consequence of the fact that *map*, *concat* and *true* form a structure that category theory calls a *monad*, and the composition operator  $\&$  is obtained from this by a standard construction called *Kleisli composition*.

We wish to show that these properties of the logic programming primitives  $\&$  and *true*, and several others regarding also  $\parallel$  and *false*, can be alternatively proved with no reference to category theory. The proofs we sketch show how a standard tool in functional programming, equational reasoning, can be applied to logic programming by means of our embedding.

All the algebraic properties we quote here can be proved equationally using only the definitions of the operators and the standard laws (see [4]) for *concat*, *map* and functional composition. As an example, given:

$$\text{map } f \cdot \text{concat} = \text{concat} \cdot \text{map } (\text{map } f), \quad (1)$$

$$\text{concat} \cdot \text{concat} = \text{concat} \cdot \text{map } \text{concat}, \quad (2)$$

$$\text{map } (f \cdot g) = (\text{map } f) \cdot (\text{map } g), \quad (3)$$

we can prove the associativity of  $\&$  by the following rewriting:

$$\begin{aligned} & (p \& q) \& r \\ &= \text{concat} \cdot \text{map } r \cdot \text{concat} \cdot \text{map } q \cdot p && \text{by defn. of } \& \\ &= \text{concat} \cdot \text{concat} \cdot \text{map } (\text{map } r) \cdot \text{map } q \cdot p && \text{by (1)} \\ &= \text{concat} \cdot \text{map } \text{concat} \cdot \text{map } (\text{map } r) \cdot \text{map } q \cdot p && \text{by (2)} \\ &= \text{concat} \cdot \text{map } (\text{concat} \cdot \text{map } r \cdot q) \cdot p && \text{by (3)} \\ &= p \& (q \& r). && \text{by defn. of } \& \end{aligned}$$

The proofs of the following properties are at least as elementary as this. The predicate *false* is a left zero for  $\&$ , but this operator is strict in its left argument, so *false* is not a right zero. This corresponds to the feature of Prolog that *false*  $\&$  *q* has that same behaviour as *false*, but *p*  $\&$  *false* may fail infinitely if *p* does. Owing to the properties of *concat* and  $\cdot$ , the  $\parallel$  operator is associative and has *false* as a left and right identity.

Other identities that are satisfied by the connectives of propositional logic are not shared by our operators because in our stream-based implementation, answers are produced in a definite order and with definite multiplicity. This behaviour mirrors the operational behaviour of Prolog. For example, the  $\parallel$  operator is not idempotent, because *true*  $\parallel$  *true* produces its input answer twice as an output, but *true* itself produces only one answer. The  $\&$  operator also fails to be idempotent, because the predicate

$$(\text{true} \parallel \text{true}) \& (\text{true} \parallel \text{true})$$

produces the same answer four times rather than just twice.

We might also expect

$$p \& (q \parallel r) = (p \& q) \parallel (p \& r),$$

that is, for  $\&$  to distribute over  $\parallel$ , but this is not the case. For a counterexample, take for *p* the predicate  $X \doteq a \parallel X \doteq b$ , for *q* the predicate  $Y \doteq c$ , and



for  $r$  the predicate  $Y \doteq d$ . Then the left-hand side of the above equation produces the four answers  $[X=a, Y=c]$ ;  $[X=a, Y=d]$ ;  $[X=b, Y=c]$ ;  $[X=b, Y=d]$  in that order, but the right-hand side produces the same answers in the order  $[X=a, Y=c]$ ;  $[X=b, Y=c]$ ;  $[X=a, Y=d]$ ;  $[X=b, Y=d]$ .

However, the other distributive law,

$$(p \parallel q) \& r = (p \& r) \parallel (q \& r),$$

does hold, and it is vitally important to the unfolding steps of program transformation. The simple proof depends on the fact that both  $\text{map } r$  and  $\text{concat}$  are homomorphisms with respect to  $++$ :

$$\begin{aligned} & ((p \parallel q) \& r) x \\ &= \text{concat } (\text{map } r (p \ x \ ++ \ q \ x)) && \text{by defn. of } \parallel, \& \\ &= \text{concat } (\text{map } r (p \ x) \ ++ \ \text{map } r (q \ x)) && \text{map} \\ &= \text{concat } (\text{map } r (p \ x)) \ ++ \ \text{concat } (\text{map } r (q \ x)) && \text{concat} \\ &= ((p \& r) \parallel (q \& r)) x. && \text{by defn. of } \& \end{aligned}$$

The declarative reading of logic programs suggests that also the following properties of  $\doteq$  and  $\exists$  ought to hold, where  $p \ x$  and  $q \ x$  are predicates and  $u$  is a term not containing  $x$ :

$$\begin{aligned} & (\exists x \rightarrow p(x) \parallel q(x)) = (\exists x \rightarrow p(x)) \parallel (\exists x \rightarrow q(x)), \\ & (\exists x \rightarrow x \doteq u \& p(x)) = p(u), \\ & (\exists x \rightarrow (\exists y \rightarrow p(x, y))) = (\exists y \rightarrow (\exists x \rightarrow p(x, y))). \end{aligned}$$

These properties are important in program transformations that manipulate quantifiers and equations, since they allow local variables to be introduced and eliminated, and allow equals to be substituted for equals in arbitrary formulas.

However, these properties of  $\doteq$  and  $\exists$  depend on properties of predicates  $p$  and  $q$  that are not shared by all functions of this type, but are shared by all predicates that are defined purely in terms of our operators. In future work, we plan to formulate precisely the ‘healthiness’ properties of definable predicates on which these transformation laws depend, such as monotonicity and substitutivity.

It might be seen as a weakness of our approach based on a ‘shallow’ embedding of Prolog in Haskell that these properties must be expressed in terms of the weak notion of a *predicate* definable in terms of our operators, when a ‘deep’ embedding (i.e., an interpreter for Prolog written in Haskell) would allow us to formulate and prove them as an inductive property of *program texts*. We believe that this is a price well worth paying for the simplicity and the clear declarative and operational semantics of our embedding.

## 5 Different Search Strategies

Our implementation of `||`, together with the laziness of Haskell, causes the search for answers to behave like depth-first search in Prolog: when computing `p x ++ q x` all the answers corresponding to the `p x` part of the search tree are returned before the other part is explored. A fair *search* strategy would share the computation effort more evenly between the two parts. Similarly, our implementation of `&` results in a left-to-right selection of the literals of a clause. A fair *selection* rule would allow one to choose the literals in a different order.

One possible solution (inspired by [10]) is to *interleave* the streams of answers, taking one answer from each stream in turn. A function *twiddle* that interleaves two lists can be defined as:

```
twiddle :: [a] -> [a] -> [a]
twiddle [] ys = ys
twiddle (x : xs) ys = x : (twiddle ys xs).
```

The operators `||` and `&` can be redefined by replacing `++` with *twiddle* and recalling that `concat = foldr (++) []`:

```
(p || q) x = twiddle (p x) (q x)
(p & q) x = foldr twiddle [] . map q . p
```

This implementation of `&` is fairer, producing in a finite time solutions of `q` that are based on later solutions returned by `p`, even if the first such solution produces an infinite stream of answers from `q`. The original implementation of `&` produces all solutions of `q` that are based on the first solution produced by `p` before producing any that are based on the second solution from `p`.

Note that this implementation of operators does *not* give breadth-first search of the search tree; it deals with infinite success but not with infinite failure. Even in the interleaved implementation, the first element of the answer list has to be computed before we can ‘switch branches’; if this takes an infinite number of steps the other branch will never be reached.

To implement breadth-first search in the embedding, the *Predicate* data-type needs to be changed. It is no longer adequate to return a single, flat stream of answers; this model is not refined enough to take into account the number of *computation steps* needed to produce a single answer. The key idea is to let *Predicate* return a stream of lists of answers, where each list represents the answers reached at the same depth, or level, of the search tree. These lists of answers with the same cost are always finite since there is only a finite number of nodes at each level of the search tree. The new type of *Predicate* is thus:

```
Predicate :: Answer -> Stream (List Answer).
```

Intuitively, each successive list of answers in the stream contains the answers with the same computational ‘cost’. The cost of an answer increases with

every resolution step in its computation. This can be captured by adding a new function *step* in the definition of predicates. For example, *append* should be coded as:

$$\begin{aligned} \text{append}(Ps, Qs, Rs) = \\ \text{step}((Ps \doteq \text{nil} \ \& \ Qs \doteq Rs) \parallel \\ (\exists X, Xs, Ys \rightarrow Ps \doteq \text{cons}(X, Xs) \ \& \ Rs \doteq \text{cons}(X, Ys) \ \& \\ \text{append}(Xs, Qs, Ys))). \end{aligned}$$

In the depth-first model, *step* is the identity function on predicates, but in the breadth-first model it is defined as follows:

$$\begin{aligned} \text{step} :: \text{Predicate} \longrightarrow \text{Predicate} \\ \text{step } p \ x = [] : (p \ x). \end{aligned}$$

Thus, in the stream returned by *step p*, there are no answers of cost 0, and for each *n*, the answers of *step p* with cost *n*+1 are the same as the answers of *p* that have cost *n*.


The implementations of the *Predicate* combinators  $\parallel$  and  $\&$  need to be changed so that they no longer operate on lists but on streams of lists. They must preserve the cost information that is embedded in the input lists. Since the cost corresponds to the level of the answer in the search tree, only resolution steps are charged for, while the applications of  $\parallel$ ,  $\&$  and *equals* are cost-free. The  $\parallel$  operator simply zips the two streams into a single one, by concatenating all the sublists of answers with the same cost. If the two streams are of different lengths, the zipping must not stop when it reaches the end of the shorter stream. We give the name *mergewith* to a specialized version of *zipwith* that has this property, and arrive at this implementation of  $\parallel$  in the breadth-first model:

$$(p \parallel q) \ x = \text{mergewith } (++) \ (p \ x) \ (q \ x).$$

The implementation of  $\&$  is harder. The cost of each of the answers to  $(p \ \& \ q)$  is a sum of the costs of the computation of *p* and the computation of *q*. The idea is first to compute all the answers, and then to flatten the resulting stream of lists of streams of lists of answers to a stream of lists of answers according to the cost. This flattening is done by the *shuffle* function which is explained below. The  $\&$ -operator is thus:

$$p \ \& \ q = \text{shuffle} \cdot \text{map} \ (\text{map } q) \cdot p$$

We write *S* for streams and *L* for finite lists for sake of brevity. The result of  $\text{map} \ (\text{map } q) \cdot p$  is of type *SLSL*. It can be visualised as a matrix of matrices, where each element of the outer matrix corresponds to a single answer of *p*. Each such answer is used as an input to *q* and consequently gives rise to a new stream of lists of answers, which are represented by the elements of the inner matrices. The rows of both the main matrix and the sub-matrices are finite, while the columns of both can be infinite. For example, the answers of  $\text{map} \ (\text{map } q) \cdot p$  with cost 2 are marked in the drawing below:



The function *shuffle* collects all the answers marked in the drawing into a single list, the third in the resulting stream of lists of answers. It is given an *SLSL* of answers, and it has to return an *SL*. Two auxiliary functions are required to do this: *diag* and *transpose*. A stream of streams is converted to a stream of lists by *diag*, and a list of streams can be converted to a stream of lists by *transpose*:

$$\begin{aligned} \text{diag} &:: \text{Stream} (\text{Stream } a) \longrightarrow \text{Stream} (\text{List } a) \\ \text{diag } xss &= [[(xss ! i) ! (n - i) \mid i \leftarrow [0..n]] \mid n \leftarrow [0..]], \end{aligned}$$

$$\begin{aligned} \text{transpose} &:: \text{List} (\text{Stream } a) \longrightarrow \text{Stream} (\text{List } a) \\ \text{transpose } xss &= \text{map } \text{hd } xss : \text{transpose} (\text{map } \text{tl } xss). \end{aligned}$$

Given *diag* and *transpose*, the function *shuffle* can be implemented as follows. The input to *shuffle* is of type *SLSL*. The application of *map transpose* swaps the middle *SL* to a *LS*, and gives *SLL*. Then the application of *diag* converts the outermost *SS* to *SL* and returns *SLLL*. This can now be used as input to *map (concat · concat)* which flattens the three innermost levels of lists into a single list, and returns *SL*:

$$\text{shuffle} = \text{map} (\text{concat} \cdot \text{concat}) \cdot \text{diag} \cdot \text{map } \text{transpose}.$$

A very interesting aspect of this breadth-first model of logic programming is that all the algebraic laws listed in the previous section still hold, if we ignore the ordering of the answers within each sublist in the main stream. This can be achieved by implementing the type of predicates as a function from answers to streams of *bags* of answers. Each of the bags contains the answers with the same computational cost, so we know that all the bags are finite. This is because there are only a finite number of branches in each node in the search tree. Hence all the equalities in our laws are still computable.

To implement *both* depth-first search and breadth-first search in the embedding, the model has to be further refined. It is not sufficient to implement predicates as functions returning streams of answer lists; they have to operate on lists of trees. The operators *||* and *&* are redefined to be operations on lists of trees, where the first one connects two lists of trees in a single one and the second ‘grafts’ trees with small subtrees at the leaves to form normal trees. If just trees

were used, rather than lists of trees,  $p \parallel q$  would have to combine their trees of answers by inserting them under a new parent node in a new tree, but that would increase the cost of each answer to  $p \parallel q$  by one. We describe this general model fully in [16].

It is interesting how concise the definitions of  $\parallel$  and  $\&$  remain in all three models. To recapitulate the three definitions of  $\&$  in the depth-first model, breadth-first model and the tree model which accommodates both search strategies, respectively:

$$\begin{aligned} p \& q &= \text{concat} \cdot \text{map } q \cdot p, \\ p \& q &= \text{shuffle} \cdot \text{map } (\text{map } q) \cdot p, \\ p \& q &= \text{graft} \cdot \text{treemap } q \cdot p. \end{aligned}$$

These closely parallel definitions hint at a deeper algebraic structure, and in fact the definitions are all instances of the so-called Kleisli construction from category theory. Even greater similarities between the three models exist, and we give a more detailed study of the relation between the three in [16].

## 6 Further Work

The work presented in this paper has not addressed the question of an efficient implementation of these ideas, although a language implementation based on our embedding is conceivable. Rather, this work is directed towards producing and using a theoretical tool (with a simple implementation) for the analysis of different aspects of logic programs. The simplicity is the key idea and the main strength of our embedding, and it has served well in opening several directions for further research.

We are presently working on two applications of the embedding. One is a study of program transformation by equational reasoning, using the algebraic laws of the embedding. The other is a categorical study of a model in which trees are used as the data-structure for the answers, and we show that there exists a *morphism of monads* between this most general model and the two models that is presented in this paper. This line of research is inspired by [17, 19].

Among other questions that we plan to address soon are also the implementation of higher-order functions and the implementation of nested functions in the embedded predicates. Constraint logic programming also has a simple model in our embedding: one has to pass equations (instead of substitutions) as parts of answers. These equations are evaluated when they become sufficiently instantiated. An efficient language implementation is also a challenging goal in this setting.

## References

- [1] Abelson and Sussman. *Structure and Interpretation of Computer Programs*, chapter 4. 1985.
- [2] M. Bellia and G. Levi. The relation between logic and functional languages: a survey. *Journal of Logic Programming*, 3(3):317–236, 1986.
- [3] R. Bird and O. de Moor. *Algebra of Programming*. Prentice Hall, 1997.
- [4] R. Bird and P. Wadler. *Introduction to Functional Programming*. Prentice Hall, 1988.
- [5] E. Giovanetti, G. Levi, C. Moiso, and C. Palamidessi. Kernel-LEAF: A logic plus functional language. *Journal of Computer and System Sciences*, 42(2), 1991.
- [6] M. Hanus. The integration of functions into logic programming: From theory to practice. *Journal of Logic Programming*, 19(20):583–628, 1994.
- [7] M. Hanus, H. Kuchen, and J.J. Moreno-Navarro. Curry: A truly functional logic language. In *Proc. ILPS'95 Workshop on Visions for the Future of Logic Programming*, pages 95–107, 1995.
- [8] C.A.R. Hoare and H. Jifeng. *Unifying Theories of Programming*. Prentice Hall, 1998.
- [9] J.W. Lloyd. Declarative programming in Escher. Technical Report CSTR-95-013, Department of Computer Science, University of Bristol, June 1995.
- [10] R. McPhee and O. de Moor. Compositional logic programming. In *Proceedings of the JICSLP'96 post-conference workshop: Multi-paradigm logic programming*, Report 96-28. Technische Universität Berlin, 1996.
- [11] Mellish and Hardy. Integrating prolog in the POPLOG environment. In J. Campbell, editor, *Implementations of Prolog*. 1984.
- [12] J. Moreno-Navarro and M. Roderiguez-Artalejo. Logic programming with functions and predicates: The language Babel. *Journal of Logic Programming*, 12(3):191–223, 1992.
- [13] L.C. Paulson. Lessons learned from LCF: a survey of natural deduction proofs. *Computer Journal*, (28), 1985.
- [14] J.A. Robinson. Beyond LogLisp: combining functional and relational programming in a reduction setting. *Machine intelligence*, 11, 1988.
- [15] J.A. Robinson and E.E. Sibert. LogLisp: An alternative to Prolog. *Machine Intelligence*, 10, 1982.

- 
- [16] S. Seres, J.M. Spivey, and C.A.R. Hoare. Algebra of logic programming. submitted to International Conference on Logic Programming, 1999.
  - [17] J.M. Spivey. A categorical approach to the theory of lists. In *Mathematics of Program Construction*. Springer LNCS 375, 1989.
  - [18] P. Wadler. How to replace failure by a list of successes. In *2'nd International Conference on Functional Programming Languages and Computer Architecture*, Nancy, France, September 1985. Springer-Verlag.
  - [19] P. Wadler. The essence of functional programming. In *19'th Annual Symposium on Principles of Programming Languages*, January 1992.